

## Paper 053-29

**Automatic logon to Mainframe and transfer SAS® datasets**

Prasad S. Ravi, Corus Pharma Inc, Seattle, WA

**ABSTRACT:**

Logging on to mainframe from PC is usually a tedious process. Tools can be developed to simplify the logging on process, one such tool is demonstrated in this paper which can be easily applied to other platforms. This paper also demonstrates an easier method of transferring SAS data sets from mainframe to PC using a SAS/CONNECT procedure.

**INTRODUCTION:**

The normal procedure to signon to remote host is to run a script file with several options and manually entering the username and password in a pop up window, which is very inconvenient to run batch or scheduled jobs. The method in this paper describes automatically logging on to mainframe without typing your username and password each time from PC for remotely submitting SAS programs on MVS. PROC DOWNLOAD procedure is used to download SAS datasets once remote signon is established to transfer data. SAS Macro language is used extensively to generalize the usage of the tool to transfer different SAS dataset files, hence this paper is written for intermediate SAS users who have good understanding of SAS macro language.

**PROCESS:**

The common commands used for signing on to mainframe from PC are as follows.

```
*filename rlink "G:\SAS8\SAS\CONNECT\SASLINK\IPHOUSECAL.SCR";
/* This is SAS Institute's version of the script file to logon to MVS */
filename rlink "H:\IPHOUSECAL.SCR";
/** This is modified version of the script file to logon to MVS **/
options comamid=tcp remote=HCSF PAGESIZE=49 LINESIZE=132;
signon HCSF;
```

Option comamid=tcp is specified to identify the access method.

And remote=hcsf option identifies the remote host.

Filename rlink runs the script file stored in the specified path.

Following is a sample script file provided by SAS Institute along with the software, in which modifications are made to three highlighted input parameters USERNAME, PASSWORD and dept. ACCT # for job submission. First identify the correct script file in SAS\CONNECT\SASLINK directory, make a copy of it into a personal drive, and open it in NOTEPAD or WORDPAD to edit, since it has .scr extension it cannot be opened in SAS program editor, then the username, password and acct# are hardcoded and saved in a personal network drive which is inaccessible to other users for security reasons.

**SCRIPT FILE:**

```
/***** Script File – IPHOUSECAL.SCR *****/

/* trace on; */

echo on;

/*-----*/
/*--      Copyright (C) 1990 by SAS Institute Inc., Cary NC      --*/
/*-- name:   iphousecal.scr                                     --*/
/*-- purpose: SAS/CONNECT SIGNON/SIGNOFF script for connecting --*/
/*--      to any TSO host via the TCP access method              --*/
```

```

/*-- notes:  1. This script may need modifications that account  --*/
/*--          for the local flavour of your MVS/TSO environment.      --*/
/*--          The logon procedure should mimic the events that      --*/
/*--          you go through when "telnet"-ing to the same          --*/
/*--          TSO host.                                           --*/
/*--          2. You must have specified OPTIONS COMAMID=TCP      --*/
/*--          in the local SAS session before using the signon      --*/
/*--          command.                                           --*/
/*-- assumes: 1. The command to execute SAS in your remote      --*/
/*--          (MVS/TSO) environment is "sas". If this is          --*/
/*--          incorrect for your site, change the contents of      --*/
/*--          the line that contains:                             --*/
/*--          type "sas ...                                         --*/
/*-- support:  SAS Institute staff                                --*/
/*-----*/

log "NOTE: Script file 'IPHOUSECAL.SCR' entered.";
if not tcp then goto notcp;
if signoff then goto signoff;
/* ----- TCP SIGNON -----*/
/* make sure we are running ibm's tcp/ip */
waitfor 'IKJ56700A ENTER', 120 seconds : noinit;
/*-----MVS LOGON-----*/
waitfor ' USERID -'      : username, 30 seconds;
nobcmmsg;
/* NOTE: IKJ56485I AND IKJ56400A STATEMENTS ADDED JUST FOR TSOXC */
/*   TO HANDLE ACCOUNT NUMBER ENTRY.                               */
waitfor 'READY',
    'ENTER CURRENT PASSWORD'          : p_enter,
    'PASSWORD NOT AUTHORIZED FOR USERID' : p_retry,
    'CURRENT PASSWORD HAS EXPIRED'      : p_expire,
    'IKJ56425I LOGON rejected, UserId'  : uronproc,
    'CURRENTLY LOGGED ON'               : dup_log,
    'NOT VALID'                        : nouser,
    'IKJ56420I'                        : wrongid,
    'PASSWORD INVALID'                 : nopass,
    'RECONNECT SUCCESS'                : recon,
    'NO BROADCAST MESSAGES'           : nobcmmsg,
    'IKJ56485I'                       : pass,
    'IKJ56484I'                       : entacct1,

```

```

        'IKJ56481I'                : entproc1,
        'IKJ56400A ENTER LOGON OR LOGOFF-'      : wrongacc,
        'IKJ56400I ENTER LOGON OR LOGOFF-'      : loffproc,
        20 seconds                    : nostrt;

strt_sas:
    log 'NOTE: Logged on to TSO.... Starting remote SAS now.';
    /* tcpipprf option-locates TCP config data sets - value is site */
    /* specific and its need dependent on your TCP configuration. */
    /* noterminal suppresses prompts from remote SAS session. */
    /* no$syntaxcheck prevents remote side from going into syntax */
    /* checking mode when a syntax error is encountered. */
    /* NOTE - THE SAS STATEMENT HERE IS DIFFERENT THAN ON IPHOUSECHI */
    /* SCRIPT. THE SALINAS BOX CAN NOT HAVE NOPREFIX SPECIFIED */
    type "SAS811 o('dmr,comamid=TCP,noterminal,no$syntaxcheck') ";
    type "autoexec('ei00.cp000000.esas.sas(connect)')" LF;
    waitfor 'SESSION ESTABLISHED', 120 seconds : nosas;
    log 'NOTE: SAS/CONNECT conversation established.';
    stop;
/*-----MVS LOGOFF-----*/

signoff:
    type 'logoff' LF;
    waitfor 'LOGGED OFF' , 20 seconds : noterm;
    log 'NOTE: SAS/CONNECT conversation terminated.';
    stop;
/*----- SUBROUTINES -----*/

recon:
    log 'NOTE: Reconnected to previous session. Old SAS session lost.';
    type "$abort" LF;
    waitfor 'READY'      : strt_sas,
        120 seconds;
    log 'NOTE: Reconnected to a Running Session, but no READY prompt';
    abort;
/*----- NEW ADDED ROUTINES -----*/
/*----- All routines are the same as the IPHOUSECHI.SCR except for */
/*----- entacct. The Salinas Box does not require an entry be made */
/*----- to this field. Added entacct1 to require employees to enter */
/*----- an Account Number. */

p_enter:
    *input nodisplay 'ENTER PASSWORD';

```

```

    type 'XXXXXX' LF;
    goto nobcmsg;
p_retry:
    log 'ERROR: Password is incorrect. Password not authorized for userid.';
    input nodisplay 'Incorrect Password. Please RE-ENTER Password';
    type LF;
    goto nobcmsg;
p_expire:
    log 'NOTE: Password has Expired. Please enter NEW password.';
    input nodisplay 'Expired Password. Please re-enter NEW Password';
    type LF;
    goto nobcmsg;
pass:
    type LF;
    goto nobcmsg;

entacct1:
    *input 'Account Number?';
    type 'XXXXXX' LF;
    *type LF;
    goto nobcmsg;
entproc1:
    log 'NOTE: @HFCIKJ is default proc name.';
    type '@hfcikj' CR LF;
    goto nobcmsg;
uronproc:
    log 'NOTE: You are already logged on.';
    type "LOGOFF" LF;
    abort;
loffproc:
    log 'NOTE: Force the logoff.';
    type "LOGOFF" LF;
    abort;
username:
    *input 'User ID? ';
    type 'XXXXXX' LF;
    *type cr;
    goto nobcmsg;
/*----- ERROR ROUTINES -----*/

```

```

nouser:
    log 'ERROR: Unrecognized userid.';
    abort;
wrongid:
    log 'ID not authorized for TSO - reenter ID.';
    goto username;
/* WRONGACC SECTION ADDED FOR TSOXC ONLY. THIS ABORTS LOGON    */
/* ATTEMPTS WHEN ACCOUNT NUMBER IS TYPED INCORRECTLY          */
wrongacc:
    log 'NOTE - VERIFY ID, PASSWORD AND ACCOUNT NUMBER FOR TSOXC';
    log 'AND TRY LOGGING ON AGAIN. THIS SESSION IS ABORTING!';
    abort;
nopass:
    log 'ERROR: Invalid password. Carefully re-enter your password.';
    goto p_enter;
notcp:
    log 'ERROR: Incorrect communications access method.';
    log 'NOTE: You must set "OPTIONS COMAMID=TCP;" before using this';
    log '    script file.';
    abort;
noinit:
    log 'ERROR: Did not understand remote session banner.';
    abort;
nostrt:
    log 'ERROR: Did not get TSO startup messages after logon.';
    abort;
nosas:
    log 'ERROR: Did not get SAS software startup messages.';
    abort;
dup_log:
    log 'ERROR: User is already logged onto TSO.';
    abort;
noterm:
    log 'WARNING: Did not get messages confirming logoff.';
    abort;
/***** END OF FILE *****/

```

After modifying the above script file with the hardcoded input parameters make the necessary site specific and software location changes where needed in the script file and save the file in a location accessed only by you. This process of creating user defined script file has to be done separately for each

user who intends to run the programs on remote session. And include the logon commands in a macro statement as follows, which can be called from other SAS programs.

```
%macro hcsf;
  *filename rlink "G:\SAS8\SAS\CONNECT\SASLINK\IPHOUSECAL.SCR";
/* This is SAS Institute's version of the script file to logon to MVS */
  filename rlink "H:\IPHOUSECAL.SCR";
/* This is modified version of the script file to logon to MVS */
  options comamid=tcp remote=HCSF PAGESIZE=49 LINESIZE=132;
  signon HCSF;
%mend hcsf;
```

### DOWNLOAD PROGRAM:

The following program signs on to remote host MVS from PC SAS and downloads a SAS dataset.

```
/******
/****      PROGRAM NAME:  DOWNLOAD.SAS      ****/
/****      LOCATION       :  R:\M-Z\PRASAD\MACROS      ****/
/****      PURPOSE        :  TO DOWNLOAD DATA FROM MAINFRAME      ****/
/****      DATE CREATED    :    06/19/01      ****/
/****      CREATED BY     :    PRASAD RAVI      ****/
/****      INPUT DATA    :    SAS DATASET ON MAINFRAME      ****/
/****      OUTPUT         :    SAS DATASET ON TO PC      ****/
/*****
/****      MODIFICATIONS:      ****/
/****      MODIFIED BY   :      ****/
/****      MODIFIED DATE:      ****/
/****      MODIFICATIONS:      ****/
/*****
options macrogen symbolgen mprint mlogic noxwait noxsync;
options mautosource mrecall sasautos=(sasautos
                                     'R:\M-Z\PRASAD\MACROS'
                                     'R:\M-Z\PRASAD\PROG');
/***** TO LOG ON TO MAINFRAME *****/
%hcsf  /**** for logging into mainframe enter your HCS USERID, PASSWORD,
        AND ACCOUNT NUMBER IS 5723E ****/
/***** TO DOWNLAOD DATA FROM MAINFRAME *****/
%macro download(mvsdb,mvsdsn,pcdir,pcdsn,where);
  options noxwait noxsync;
/* Creating a sub-dir under PC dir using the mainframe dataset name */
  libname pc "&pcdir.\&mvsdb";
  %if %sysfunc(libref(pc)) ne 0 %then %do;
    x "mkdir &pcdir.\&mvsdb";
  %end;
/**** Creating a filename to send the log of the run to PC directory ****/
  filename downlog "&pcdir.\&mvsdb.\DOWNLOAD_&mvsdb..log";
  %lput(mvsdata,&mvsdb)
  %lput(dsn1,&mvsdsn)
  %lput(dsn2,&pcdsn)
  %lput(wherecl,&where)
  proc printto new log=downlog;
  run;
  libname pc "&pcdir.\&mvsdb";
  rsubmit;
  LIBNAME USER CLEAR;
  LIBNAME USER 'WORK' SPACE=(CYL,(200,100));
  options nos99nomig;
  libname mvs "&mvsdata" disp=shr;
  proc download status=n data=mvs.&dsn1 out=pc.&dsn2;
    &wherecl;
```

```

                                run;
                                libname mvs;
                                endrsubmit;
                                libname pc;
                                proc printto;
                                run;
                                filename downlog;
                                %mend download;

%download(%str(HCSIW8.SAS.ANALYTIC),
          STRAT,
          %str(R:\M-Z\PRASAD\DATA\ADHOC\DATA),
          STRAT,
          %str())
signoff;
/***** END OF THE PROGRAM *****/

```

### DESCRIPTION OF THE DOWNLOAD MACRO COMPONENTS:

SASAUTOS option is used to access the compiled following macros from a macro library.

%hcsf macro contains the logon commands to signon to remote host MVS from local host PC.

%lput macro transfers macro variable values filenames and dataset names from local host to remote host.

Proc printto is used to save the log to check for the syntax errors and successful transfer of the datasets.

Libnames for PC and MVS environments are specified to identify the paths of the directory and files to be transferred to and from local and remote hosts.

Libname user is defined to allocate the SAS work space in the remote host.

Proc download procedure downloads the SAS dataset from remote host MVS to PC local host,

STATUS=N option in proc download suppresses the display of download status window.

A where clause macro variable is provided to subset the dataset if needed.

A signoff statement is necessary at the end of the program to signoff from the remote host session.

### CONCLUSION:

This paper illustrates an easier and automated process of logging on to mainframe by modifying the logon script file from PC SAS environment to remotely submit SAS jobs on MVS and also downloading SAS datasets as quickly as possible using a SAS/CONNECT procedure.

### CONTACT INFORMATION:

Your comments and questions are valued and encouraged. Contact the author at:

Prasad Ravi

Corus Pharma Inc

2025 First Avenue, Suite 800

Seattle, WA 98121

Work Phone: 206-728-5090

Fax: 206-728-5095

Email: pravi@coruspharma.com

Web: www.coruspharma.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.